How to set forward voltage for a diode model in NGSpice?
You do not simply "set" the forward voltage. The forward voltage is determined by a combination of model parameters. Here's a model I found for the 1N4148 which is a commonly used Silicon diode:

******************************************
*1N4148
*VRRM = 100V
*IFRM = 450 mA 
*trr  = 4ns
*
*Package: SOD 27
*
*Package Pin 1 : Cathode
*Package Pin 2 : Anode
*
*Simulator: PSPICE
*
******************************************
*
.SUBCKT 1N4148 1 2 
*
* The resistor R1 does not reflect 
* a physical device. Instead it
* improves modeling in the reverse 
* mode of operation.
*
R1 1 2 5.827E+9 
D1 1 2 1N4148
*
.MODEL 1N4148 D 
+ IS = 4.352E-9 
+ N = 1.906 
+ BV = 110 
+ IBV = 0.0001 
+ RS = 0.6458 
+ CJO = 7.048E-13 
+ VJ = 0.869 
+ M = 0.03 
+ FC = 0.5 
+ TT = 3.48E-9 
.ENDS


If you use these parameters with the standard PSPICE diode model you should get (I did not check it for you, so no guarantees) a diode with a forward voltage of around 0.7 V at "normal" forward operation (a couple of mA of current flowing).